

Module 24 Partiële differentiaalvergelijkingen

Onderwerp	Partiële differentiaalvergelijkingen
Expressies	pdsolve, dchange, animate, build
Bibliotheken	PDEtools, plots
Zie ook	Module 21.

24.1 Eerste orde lineaire partiële differentiaalvergelijkingen

De algemene vorm van een eerste orde lineaire partiële differentiaalvergelijking in het vlak is

$$a(x, y) u_x + b(x, y) u_y + c(x, y) u = f(x, y), \quad (24.1)$$

met u_x en u_y de partiële afgeleiden van de functie u naar x en y respectievelijk. We kunnen de oplossing van (24.1) bepalen langs *karakteristieken*. De karakteristieken zijn oplossingskrommen van het stelsel gewone differentiaalvergelijkingen

$$\begin{cases} \frac{dx}{ds} = a(x, y) \\ \frac{dy}{ds} = b(x, y) \end{cases} \quad (24.2)$$

Als we schrijven $U(s) = u(x(s), y(s))$, dan wordt langs karakteristieken U bepaald door op te lossen

$$\frac{dU}{ds} = -\tilde{c}(s) U + \tilde{f}(s) \quad (24.3)$$

waarin $\tilde{c}(s) = c(x(s), y(s))$, en $\tilde{f}(s) = f(x(s), y(s))$.

24.2 Oplossing langs karakteristieken

Met Maple kunnen we zowel het richtingsveld voor de karakteristieken tekenen, als ook oplossingen van (24.3) al dan niet numeriek bepalen.

Voorbeeldopgave

Bepaal de karakteristieken voor de vergelijking

$$y u_x - x u_y + u = 0.$$

Neem vervolgens aan dat u wordt voorgeschreven op de positieve x -as door $u(x, 0) = \phi(x)$ en laat zien dat de oplossing dan wordt gegeven door:

$$u(x, y) = \phi\left(\sqrt{x^2 + y^2}\right) \exp(-\arctan(-y/x)).$$

Voorbeeldsessie

```
> restart;
Algemene oplossing van (24.2):
> dsolve( {diff(x(s),s)=y(s),diff(y(s),s)=-x(s)}, {x(s),y(s)} );
      {x(s) = _C1 sin(s) + _C2 cos(s), y(s) = _C1 cos(s) - _C2 sin(s)}
Idem door het punt (x0, 0) :
> sol :=
  dsolve( {diff(x(s),s)=y(s),diff(y(s),s)=-x(s),x(0)=x0,y(0)=0},
    {x(s),y(s)} );
      sol := {y(s) = -x0 sin(s), x(s) = x0 cos(s)}
> XY := subs( sol, {x=x(s), y=y(s)} );
      XY := {x = x0 cos(s), y = -x0 sin(s)}
Oplossing van (24.3) door het punt U(0) = phi(x0) :
> dsolve( {diff(U(s),s)=-U(s),U(0)=phi(x0)}, U(s) );
      U(s) = phi(x0) e^(-s)
> solve( XY, {x0,s} ): allvalues(%);
```

$$\left\{ \begin{aligned} x_0 &= \sqrt{y^2 + x^2}, s = \arctan\left(-\frac{y}{\sqrt{y^2 + x^2}}, \frac{x}{\sqrt{y^2 + x^2}}\right) \end{aligned} \right\},$$

$$\left\{ \begin{aligned} s &= \arctan\left(\frac{y}{\sqrt{y^2 + x^2}}, -\frac{x}{\sqrt{y^2 + x^2}}\right), x_0 = -\sqrt{y^2 + x^2} \end{aligned} \right\}$$

Toelichting

Voor deze opgave is in vergelijking (24.1):

$$a(x, y) = y, \quad b(x, y) = -x, \quad c(x, y) = 1 \quad \text{en} \quad f(x, y) = 0.$$

Allereerst hebben we de algemene vorm van de karakteristieken uitgerekend. Dit blijken cirkels te zijn (met straal $\sqrt{C1^2 + C2^2}$). Vervolgens hebben we de karakteristiek bepaald die gaat door het punt $(x_0, 0)$. Dat is natuurlijk wederom de cirkel met straal $|x_0|$. Het is van belang dat we een gegeven punt in het vlak kunnen verbinden met de positieve x -as (waar de oplossing u de voorgeschreven oplossing ϕ moet hebben). Zoals Maple terecht vindt, kan dit op twee manieren. Zie §3.6 (blz. 32) voor de arctan-functie met twee argumenten. \diamond

`pdsolve`

Met het commando `pdsolve` kan zo'n oplossing trouwens soms ook direct worden bepaald.

Voorbeeldsessie

```
> restart;
> PDE := y*diff(u(x,y),x) - x*diff(u(x,y),y) + u(x,y) = 0;
      PDE := y (∂/∂x u(x, y)) - x (∂/∂y u(x, y)) + u(x, y) = 0
> pdsolve(PDE);
```

$$u(x, y) = _F1(x^2 + y^2) e^{-\arctan(\frac{x}{y})}$$

Toelichting

De functie $_F1$ is een willekeurige functie. Merk op dat $\arctan(x/y)$ en $\arctan(-y/x)$ op de constante $\frac{\pi}{2}$ na aan elkaar gelijk zijn. \diamond

24.3 Reductie tot een gewone differentiaalvergelijking

We kunnen van de karakteristieken gebruikmaken om de vergelijking op een eenvoudiger vorm te brengen. Als voorbeeld nemen we de vergelijking

$$x^2 u_x + y u_y + x y u = 1 \tag{24.4}$$

Als we nu het stelsel (24.2) bepalen dan vinden we dat

$$\frac{dy}{dx} = \frac{\frac{dy}{ds}}{\frac{dx}{ds}} = \frac{y}{x^2} \tag{24.5}$$

Integratie levert dat $\ln(y) + \frac{1}{x} = \eta$. Voor iedere waarde van de constante η hebben we een karakteristiek van de vergelijking. We gebruiken deze 'constante' als nieuwe variabele: we schrijven $u(x, y) = v(\xi, \eta)$, met η de integratieconstante. Wat we voor ξ kiezen is willekeurig, zolang het maar onafhankelijk is van η . In onderstaande Maple-sessie nemen we $\xi = x$. Na transformatie blijft een gewone differentiaalvergelijking over:

$$v_\xi + \frac{1}{\xi} \exp\left(\eta - \frac{1}{\xi}\right) v = \frac{1}{\xi^2}. \tag{24.6}$$

Voorbeeldopgave

Transformeer (24.4) via de coördinatentransformatie

$$\xi = x, \quad \eta = \ln(y) + \frac{1}{x}.$$

Voorbeeldsessie

```

> restart;
> xi,eta := x, ln(y)+1/x;
      xi, eta := x, ln(y) + 1/x
> D[1](u):=diff(v(xi,eta),x);
D[2](u):=diff(v(xi,eta),y);
      D1(u) := D1(v) ( x, ln(y) + 1/x ) - D2(v) ( x, ln(y) + 1/x )
      D2(u) := D2(v)(x, ln(y) + 1/x)
      D1(u) := D1(v) ( x, ln(y) + 1/x ) - D2(v) ( x, ln(y) + 1/x )
      D2(u) := D2(v)(x, ln(y) + 1/x)
> pde := simplify( x^2*D[1](u) + y*D[2](u) + x*y*u = 1 );
      pde := x ( D1(v)(x, ln(y) x + 1) x + y u ) = 1
> xi := 'xi': eta := 'eta':
x := xi; y := exp(-1/x)*exp(eta);
      x := xi
      y := e^(-1/x) e^eta
> subs(u=v(xi,eta),pde): simplify(%) assuming real:
ode := convert(%,diff);
      ode := xi ( ( d/dxi v(xi, eta) ) xi + v(xi, eta) e^(-1+eta/x) ) = 1

```

Toelichting

Voor iedere waarde van η hebben we een karakteristiek waarop we een gewone differentiaalvergelijking op moeten lossen. Als we de oplossing willen voorschrijven op een curve, dan moet die curve de karakteristieken snijden. De voorgeschreven waarde op de karakteristiek levert dan de beginconditie voor de gewone differentiaalvergelijking. Zodra er meer dan één snijpunt is met een karakteristiek, dan is er in het algemeen geen oplossing meer.

`convert, diff`

Met `convert(%,diff)` worden uitdrukkingen als $D[1](v)(\xi, \eta)$ herschreven als $\frac{\partial}{\partial \xi} v(\xi, \eta)$. \diamond

`dchange`

Voor coördinatentransformaties kan ook de procedure `dchange` uit de bibliotheek `PDEtools` worden gebruikt.

Voorbeeldsessie

```

> restart; with(PDEtools):
> PDV := x^2*diff(u(x,y),x) + y*diff(u(x,y),y) + x*y*u(x,y) = 1;

```

```

PDV := x^2 (∂/∂x u(x, y)) + y (∂/∂y u(x, y)) + x y u(x, y) = 1
> tr := solve( {xi=x, eta=ln(y) + 1/x}, {x,y} );
tr := { x = ξ, y = e^(ξ-1) }
> dchange(tr,PDV): simplify(%);
ξ ( (∂/∂ξ u(η, ξ)) ξ + e^(ξ-1) u(η, ξ) ) = 1

```

24.4 Canonieke vorm van tweede orde lineaire partiële differentiaalvergelijkingen

De algemene vorm van een tweede orde lineaire partiële differentiaalvergelijking wordt gegeven door:

$$Au_{xx} + 2Bu_{xy} + Cu_{yy} + Du_x + Eu_y + Fu + G = 0, \quad (24.7)$$

met A-G functies van x en y . De discriminant wordt gegeven door

$$\Delta = B^2 - AC. \quad (24.8)$$

Als $\Delta > 0$, het *hyperbolische* geval, dan worden nieuwe coördinaten gedefinieerd die constant zijn langs karakteristieken gegeven door de differentiaalvergelijkingen

$$\frac{dy}{dx} = \frac{B \pm \sqrt{B^2 - AC}}{A}. \quad (24.9)$$

Als $\Delta = 0$, het *parabolische* geval, dan wordt een nieuwe coördinaat gedefinieerd die constant is langs de karakteristiek gegeven door de differentiaalvergelijking

$$\frac{dy}{dx} = \frac{B}{A}. \quad (24.10)$$

Als $\Delta < 0$, het *elliptische* geval, dan worden nieuwe coördinaten gedefinieerd die constant zijn langs karakteristieken gegeven door de differentiaalvergelijking

$$\frac{dy}{dx} = \frac{B + i\sqrt{AC - B^2}}{A}. \quad (24.11)$$

De vergelijking is complex, en heeft één complexe integratieconstante. Het reële en het imaginaire deel leveren de twee nieuwe coördinaten.

De *normaalvormen* voor het tweede orde deel worden gegeven door

hyperbolisch	$u_{\xi\eta}$
parabolisch	$u_{\xi\xi}$
elliptisch	$u_{\xi\xi} + u_{\eta\eta}$

24.5 Voorbeelden

We geven voor elk van de drie gevallen een voorbeeld.

Voorbeeldopgave

Bepaal een coördinatentransformatie waarmee de hyperbolische differentiaalvergelijking

$$u_{xx} + 2 \cos(x) u_{xy} - \sin^2(x) u_{yy} - \sin(x) u_y = 0$$

op normaalvorm wordt gebracht.

Voorbeeldsessie

```
> restart: with(PDEtools):
> A := 1: B := cos(x): C := -sin(x)^2:
> PDV := A*diff(u(x,y),x$2) + 2*B*diff(u(x,y),x,y) + C*diff(u(x,y),y,y)
- sin(x)*diff(u(x,y),y)=0;
```

$$PDV := \frac{\partial^2}{\partial x^2} u(x, y) + 2 \cos(x) \left(\frac{\partial^2}{\partial y \partial x} u(x, y) \right) - \sin(x)^2 \left(\frac{\partial^2}{\partial y^2} u(x, y) \right) - \sin(x) \left(\frac{\partial}{\partial y} u(x, y) \right) = 0$$

Karakteristieken:

```
> dydx1 := diff(y(x),x) = (B+sqrt(B^2-A*C))/A;
```

$$dydx1 := \frac{d}{dx} y(x) = \cos(x) + \sqrt{\cos(x)^2 + \sin(x)^2}$$

```
> K1 := dsolve( dydx1, y(x));
```

$$K1 := y(x) = \sin(x) + x + _C1$$

```
> dydx2 := diff(y(x),x) = (B-sqrt(B^2-A*C))/A;
```

```
> K2 := dsolve( dydx2, y(x) );
```

$$K2 := y(x) = \sin(x) - x + _C1$$

De transformatie wordt dus:

```
> tr_inv := {xi=subs(K1,y-y(x) ), eta=subs(K2,y-y(x))};
```

$$tr_inv := \{ \xi = y - \sin(x) - x - _C1, \eta = y - \sin(x) + x - _C1 \}$$

```
> \_C1 := 0;
```

```
> tr := solve( tr_inv, {x,y} );
```

$$tr := \left\{ x = \frac{\eta}{2} - \frac{\xi}{2}, y = \frac{\xi}{2} - \sin\left(-\frac{\eta}{2} + \frac{\xi}{2}\right) + \frac{\eta}{2} \right\}$$

```
> dchange(tr,PDV): simplify(%);
```

$$-4 \left(\frac{\partial^2}{\partial \xi \partial \eta} u(\eta, \xi) \right) = 0$$

Toelichting

De vergelijking reduceert inderdaad tot de standaardvorm van een hyperbolische vergelijking. \diamond

Voorbeeldopgave

Bepaal een coördinatentransformatie waarmee de parabolische differentiaalvergelijking

$$9u_{xx} + 12u_{xy} + 4u_{yy} + u_x = 0$$

op normaalvorm wordt gebracht.

Voorbeeldsessie

```
> restart: with(PDEtools):
> A := 9: B := 6: C := 4:
> PDV:=A*diff(u(x,y),x$2)+2*B*diff(u(x,y),x,y)+C*diff(u(x,y),y,y)
+diff(u(x,y),x)=0;
PDV := 9  $\left(\frac{\partial^2}{\partial x^2} u(x, y)\right) + 12 \left(\frac{\partial^2}{\partial y \partial x} u(x, y)\right) + 4 \left(\frac{\partial^2}{\partial y^2} u(x, y)\right) + \frac{\partial}{\partial x} u(x, y) = 0$ 
> Delta:=B^2-A*C;

$$\Delta := 0$$

> K := dsolve(diff(y(x),x)=B/A,y(x));

$$K := y(x) = \frac{2x}{3} + \_C1$$

> _C1 := 0:
> tr_inv := {xi=subs(K,y-y(x)), eta=x};

$$tr\_inv := \left\{ \eta = x, \xi = y - \frac{2x}{3} \right\}$$

> tr := solve( tr_inv, {x,y} );

$$tr := \left\{ x = \eta, y = \xi + \frac{2\eta}{3} \right\}$$

> dchange(tr,PDV): simplify(%);

$$9 \left( \frac{\partial^2}{\partial \eta^2} u(\eta, \xi) \right) + \frac{\partial}{\partial \eta} u(\eta, \xi) - \frac{2}{3} \frac{\partial}{\partial \xi} u(\eta, \xi) = 0$$

```

Toelichting

Merk op dat de keuze van de tweede coördinaat, in dit geval η , willekeurig is, zolang maar de Jacobiaan van de transformatie ongelijk is aan 0. \diamond

Voorbeeldopgave

Bepaal een coördinatentransformatie waarmee de elliptische differentiaalvergelijking

$$u_{xx} + 2u_{xy} + 3u_{yy} + 4u = 0$$

op normaalvorm wordt gebracht.

Voorbeeldsessie

```
> restart: with(PDEtools):
> A := 1: B := 1: C := 3:
> PDV := A*dif(u(x,y),x$2) + 2*B*dif(u(x,y),x,y) +
      C*dif(u(x,y),y,y) + 4*u(x,y)=0;

      PDV :=  $\frac{\partial^2}{\partial x^2} u(x, y) + 2 \left( \frac{\partial^2}{\partial y \partial x} u(x, y) \right) + 3 \left( \frac{\partial^2}{\partial y^2} u(x, y) \right) + 4 u(x, y) = 0$ 
> Delta:=B^2-A*C;

      Delta := -2
> K := dsolve(dif(y(x),x)=B/A+I*sqrt(A*C-B^2)/A,y(x));

      K :=  $y(x) = x + I x \sqrt{2} + \_C1$ 
> _C1 := 0:
> use RealDomain in
      tr_inv := { xi = Re( subs(K,y-y(x)) ),
                eta = Im( subs(K,y-y(x)) ) }
end use;

      tr_inv :=  $\left\{ \xi = y - x, \eta = -x \sqrt{2} \right\}$ 
> tr := solve( tr_inv, {x,y} );

      tr :=  $\left\{ x = -\frac{\eta \sqrt{2}}{2}, y = \xi - \frac{\eta \sqrt{2}}{2} \right\}$ 
> dchange(tr,PDV): simplify(%)/2;

       $\frac{\partial^2}{\partial \eta^2} u(\eta, \xi) + \frac{\partial^2}{\partial \xi^2} u(\eta, \xi) + 2 u(\eta, \xi) = 0$ 
```

Toelichting

We hebben hier `RealDomain` gebruikt om aan te geven dat alle gebruikte variabelen reëel geacht worden te zijn. \diamond

24.6 De golfvergelijking

De oplossing van de *golfvergelijking*

$$u_{tt} = c^2 u_{xx}, \quad x \in \mathbb{R}, \quad t \in \mathbb{R}, \quad (24.12)$$

met als begincondities op $t = 0$,

$$\begin{aligned} u(x, 0) &= \varphi(x) \\ u_t(x, 0) &= \psi(x) \end{aligned} \quad (24.13)$$

wordt gegeven door de formule van d'Alembert:


$$u(x, t) = \frac{1}{2} (\varphi(x - ct) + \varphi(x + ct)) + \frac{1}{2c} \int_{x-ct}^{x+ct} \psi(\tau) d\tau. \quad (24.14)$$

Voorbeeldopgave

Laat een animatie zien van de oplossing van de golfvergelijking met $\varphi(x) = \cos(x)$, als $|x| \leq \pi/2$ en 0 elders, en $\psi \equiv 0$, en met golfsnelheid $c = 7$ op het domein $-10 \leq x \leq 10$, en $0 \leq t \leq 1$.

Voorbeeldsessie

```
> restart; with(plots):
> PDV := diff(u(x,t),t$2) = c^2*diff(u(x,t),x$2);
      PDV :=  $\frac{\partial^2}{\partial t^2} u(x, t) = c^2 \frac{\partial^2}{\partial x^2} u(x, t)$ 
> pdsolve(PDV);
      u(x, t) = _F1(ct + x) + _F2(ct - x)
> phi := x -> piecewise( abs(x)<=Pi/2, cos(x), 0 ):
> pdsolve( {PDV, u(x,0)=phi(x), D[2](u)(x,0)=0}, {u(x,t)} );
Error, (in casesplit/K) this version of casesplit is not yet handling the function:
piecewise
We proberen een beginwaarde zonder piecewise:
> s := pdsolve( {PDV, u(x,0)=cos(x), D[2](u)(x,0)=0}, {u(x,t)} );
      s :=
> c := 7:
> u := (x,t) -> 1/2*(phi(x-c*t) + phi(x+c*t));
      u := (x, t)  $\mapsto$   $1/2 \phi(x - ct) + 1/2 \phi(ct + x)$ 
> animate( u(x,t), x=-10..10, t=0..1, numpoints=250, frames=50 );
(plaatje niet getoond)
Inplaats daarvan: De toestand opt = 0 en opt = 1 :
> plot( u(x,0), x=-10..10, color=black,
      thickness=3, title="t = 0",
      tickmarks=[[ -10, -5, 0, 5, 10 ], [ 0.5, 1 ] ] );
> plot( u(x,1), x=-10..10, u=0..1, color=black,
      thickness=3, title="t = 1",
      tickmarks=[[ -10, -5, 0, 5, 10 ], [ 0.5, 1 ] ] );
(zie figuur 48)
```



FIGUUR 48. Oplossing golfvergelijking

Toelichting

Voor de aardigheid hebben we eerst even geprobeerd of Maple de golfvergelijking kan oplossen. De algemene oplossing gaat goed, maar om de oplossing van het beginwaardeprobleem te vinden is `pdsolve` blijkbaar niet bedoeld. In eerste instantie suggereert de foutmelding dat de `piecewise`-functie (nog) niet voor de begin- of randvoorwaarden gebruikt mag worden, maar als we de ‘hele’ cosinusfunctie als beginconditie nemen, komt er helemaal geen antwoord. Afwachten maar wat Maple 15 hiervan maakt; er wordt blijkbaar aan gewerkt.

We hebben in de voorbeeldsessie dus alleen plaatjes getekend van de d’Alembert-oplossing (24.14). Zie §9.4 (blz. 129) voor het plot-commando `animate`.

In dit voorbeeld zien we mooi de twee karakteristieken waarlangs, eerlijk verdeeld, de begingolf zich verplaatst. \diamond

De golfvergelijking op een halflijn. We bekijken het begin- en randwaardeprobleem

$$u_{tt} = c^2 u_{xx}, \quad x \geq 0, \quad t \in \mathbb{R}, \quad (24.15)$$

met als begincondities op $t = 0$,

$$\begin{aligned} u(x, 0) &= \varphi(x), & x \geq 0 \\ u_t(x, 0) &= \psi(x), & x \geq 0 \end{aligned} \quad (24.16)$$

en met de randwaarde:

$$u(0, t) = 0, \quad t \in \mathbb{R}. \quad (24.17)$$

De oplossing van dit probleem wordt gegeven door de formule

$$u(x, t) = \frac{1}{2} (\Phi(x - ct) + \Phi(x + ct)) + \frac{1}{2c} \int_{x-ct}^{x+ct} \Psi(\tau) d\tau, \quad (24.18)$$

met hierin

$$\Phi(x) = \begin{cases} \varphi(x) & \text{als } x \geq 0 \\ -\varphi(-x) & \text{als } x < 0 \end{cases}$$

$$\Psi(x) = \begin{cases} \psi(x) & \text{als } x \geq 0 \\ -\psi(-x) & \text{als } x < 0, \end{cases}$$

de oneven uitbreiding van φ , respectievelijk ψ tot \mathbb{R} .

Voorbeeldopgave

Maak een animatie van de oplossing van (24.15) als gegeven is dat

$$\varphi(x) = \begin{cases} \sin(x) & \text{als } 0 \leq x \leq \pi \\ 0 & \text{als } x > \pi \end{cases}$$

$\psi \equiv 0$, en $c = 7$ op het domein $0 \leq t \leq 1$, en $0 \leq x \leq 10$.

Voorbeeldsessie

```
> restart; with(plots):
> phi := x -> piecewise( x<=Pi, sin(x), 0 ):
Uitbreiden tot een oneven functie:
> Phi := x -> piecewise( x>=0, phi(x), -phi(-x) ):
> simplify(Phi(x));
```


$$\begin{cases} 0 & x < -\pi \\ \sin(x) & x \leq \pi \\ 0 & \pi < x \end{cases}$$

```
> c := 7:
> u := (x,t) -> 1/2*(Phi(x-c*t) + Phi(x+c*t));
u := (x, t) -> 1/2 * Phi(x - ct) + 1/2 * Phi(x + ct)
> animate( u(x,t), x=0..10, t=0..1, frames=50 ):
(Plaatje niet getoond)
> plot( u(x,0), x=-0..10, u=-0.5..1, color=black,
        thickness=3, title="t = 0",
        tickmarks=[[2,4,6,8,10],[-0.5,0,0.5,1]] );
> plot( u(x,1), x=0..10, u=-0.5..1, color=black,
        thickness=3, title="t = 1",
        tickmarks=[[2,4,6,8,10],[-0.5,0,0.5,1]] );
```

(Zie figuur 49.)

Toelichting

Als u de x op het interval $[-10, 10]$ inplaats van $[0, 10]$ had genomen, maakt de animatie duidelijk waarom dit wel de methode van spiegelen wordt genoemd. \diamond



FIGUUR 49. Oplossing golfvergelijking op een halflijn

De inhomogene golfvergelijking. De oplossing van het inhomogene probleem

$$\begin{aligned} u_{tt} &= c^2 u_{xx} + P(x, t), & x \in \mathbb{R}, \quad t > 0 \\ u(x, 0) &= \varphi(x), \quad u_t(x, 0) = \psi(x), & x \in \mathbb{R} \end{aligned} \tag{24.19}$$

wordt gegeven door de formule

$$\begin{aligned} u(x, t) &= \frac{1}{2}(\varphi(x - ct) + \varphi(x + ct)) + \frac{1}{2c} \int_{x-ct}^{x+ct} \psi(\tau) d\tau + \\ &\quad \frac{1}{2c} \int_0^t d\eta \int_{c\eta+x-ct}^{-c\eta+x+ct} P(\xi, \eta) d\xi \end{aligned} \tag{24.20}$$

De inhomogene term wordt dus geïntegreerd over de driehoek die ontstaat door vanuit het punt (x, t) terug in de tijd de twee karakteristieken $\xi = x \pm c(\eta - t)$ te volgen tot het snijpunt met de ξ -as.

Voorbeeldopgave

Maak een animatie van de inhomogene golfvergelijking als gegeven is dat $c = 1$, $\psi \equiv 0$, $\varphi(x) = \cos(x)$ als $|x| \leq \pi/2$, en 0 elders, en $P(x, t) = \sin(t) \sin(x)$. Neem als domein $-4 \leq x \leq 4$, $0 \leq t \leq 1$.

Voorbeeldsessie


```
> restart: with(plots):
> phi := x -> piecewise( abs(x)<=evalf(Pi/2), cos(x), 0 ):
P := (x,t) -> sin(t)*sin(x):
c := 1:
> u := (x,t) -> 1/2*(phi(x-c*t) + phi(x+c*t)) +
1/(2*c)*evalf( Int( Int(P(xi,eta),
xi=c*eta+x-c*t..-c*eta+x+c*t), eta=0..t ) ):
> u(x,0);
      { cos(x) |x| ≤ 1.5708
      { 0      otherwise
> animate( u(x,t), x=-4..4, t=0..1, frames=10,
thickness=3, color=black ):
```

(plaatje niet getoond)

Inplaats daarvan:

```
> plot( u(x,0), x=-4..4, u=-0.2..1, color=black, thickness=3,
        tickmarks=[[-4,-2,0,2,4],[0.5,1]], title="t = 0" );
> plot( u(x,1), x=-4..4, u=-0.2..1, color=black, thickness=3,
        tickmarks=[[-4,-2,0,2,4],[0.5,1]], title="t = 1" );
```

(zie figuur 50.)



FIGUUR 50. Oplossing inhomogene golfvergelijking

Toelichting

Om de dubbele integratie uit te voeren kunnen we Maple numerieke routines laten gebruiken. Dit wordt hier bewerkstelligd door de een `evalf` van de *inerte* integraal (dus `Int` inplaats van `int`) te vragen. Zie voor meer informatie: `?evalfint`, vooral van belang als je Maple een bepaald numeriek algoritme wil voorschrijven. \diamond

24.7 Scheiding van variabelen

We bekijken de homogene *warmtevergelijking* met *Dirichlet randvoorwaarden*

$$\begin{aligned} u_t &= ku_{xx} & 0 < x < L, \quad t > 0 \\ u(0, t) &= u(L, t) = 0 & t > 0 \\ u(x, 0) &= f(x) & 0 \leq x \leq L. \end{aligned} \tag{24.21}$$

We bepalen basisfuncties die voldoen aan de partiële differentiaalvergelijking inclusief de randvoorwaarden, maar nog niet aan de beginvoorwaarde, van de vorm

$$u(x, t) = X(x)T(t).$$

Er geldt dan dat

$$\frac{X''}{X} = \frac{\dot{T}}{kT} = -\lambda \quad (24.22)$$

Dit heeft voor X de oplossingen $X_n(x) = \sin \frac{n\pi x}{L}$, met bijbehorende eigenwaarde $\lambda_n = \frac{n^2\pi^2}{L^2}$. De oplossing is dan van de vorm

$$u(x, t) = \sum_{n=1}^{\infty} b_n e^{-k \frac{n^2\pi^2}{L^2} t} \sin \frac{n\pi x}{L}. \quad (24.23)$$

Om aan de beginconditie te voldoen moet er gelden dat

$$b_n = \frac{2}{L} \int_0^L f(s) \sin \frac{n\pi s}{L} ds$$

Maple gebruiken we om deze coëfficiënten te bepalen en een benadering van de oplossing (eindig veel termen uit de oneindige som) te tekenen.

Voorbeeldopgave

Los (24.21) op als gegeven is dat

$$f(x) = \begin{cases} x & \text{voor } 0 \leq x \leq \frac{L}{2} \\ L - x & \text{voor } \frac{L}{2} \leq x \leq L. \end{cases}$$

Voorbeeldsessie

```
> restart: with(plots):
> f := x -> piecewise(x<0, 0, x<=L/2, x, x<L, L-x, 0): f(x);

      \begin{cases} 0 & x < 0 \\ x & x \leq \frac{L}{2} \\ L-x & x < L \\ 0 & \text{otherwise} \end{cases}

> b := n -> 2/L*int(f(s)*sin(n*Pi*s/L),s=0..L);

      b := n -> \frac{2}{L} \int_0^L f(s) \sin\left(\frac{n\pi s}{L}\right) ds

> add( b(n)*exp(-n^2*Pi^2*k*t/L^2)*sin(n*Pi*x/L), n=1..3 ):
u := simplify(%) assuming L>0;


      u := \frac{4}{9} \frac{L e^{-\frac{\pi^2 k t}{L^2}} \left( 9 \sin\left(\frac{\pi x}{L}\right) - e^{-\frac{8\pi^2 k t}{L^2}} \sin\left(\frac{3\pi x}{L}\right) \right)}{\pi^2}

> eval( u, {L=Pi,k=1} ): U := unapply(%,x,t);

      U := (x, t) -> \frac{4}{9} \frac{e^{-t} (9 \sin(x) - e^{-8t} \sin(3x))}{\pi}

> animate( U(x,t), x=0..Pi, t=0..3, frames=100);
> plot( [seq(U(x,t), t=0..3, 0.5)], x=0..Pi, color=black,
        thickness=3 );
```

(zie figuur 51.)



FIGUUR 51. Eerste twee Fourier-termen van de oplossing van de warmtevergelijking, voor verschillende waarden van t .

Toelichting

We kijken hier naar een benadering van de oplossing met twee Fourier-termen. Het is nu niet moeilijk om meer termen mee te nemen en te bezien of dat er veel toe doet. \diamond

Aan de hand van de warmtevergelijking (24.21) met Dirichlet randvoorwaarden, bespreken we hier nog een tweetal mogelijkheden. In de eerste plaats kan Maple ons behulpzaam zijn bij het scheiden van variabelen. Het doet dat (vaak) vanzelf als we vragen om de algemene oplossing.

Voorbeeldsessie

```
> restart; with(plots):
> PDV := diff(u(x,t),t) = k*diff(u(x,t),x$2);
```

$$PDV := \frac{\partial}{\partial t} u(x, t) = k \left(\frac{\partial^2}{\partial x^2} u(x, t) \right)$$

Algemene oplossing:

```
> sol := pdsolve(PDV);
```

$$\text{sol} := (u(x, t) = _F1(x) _F2(t)) \ \&\text{where} \\ \left[\left\{ \frac{d}{dt} _F2(t) = k _c1 _F2(t), \frac{d^2}{dx^2} _F1(x) = -c1 _F1(x) \right\} \right]$$

```
> with(PDEtools):
```

```
> Sol := subs( {_c[1]=-lambda}, sol );
```

```
> build(Sol);
```

$$u(x, t) = \frac{-C3 _C1 \sin(\sqrt{\lambda} x)}{e^{k \lambda t}} + \frac{-C3 _C2 \cos(\sqrt{\lambda} x)}{e^{k \lambda t}}$$

Toelichting

Het resultaat van `pdsolve` is precies formule (24.22): de algemene oplossing is het product van een functie $_F1$ van x alleen, en een functie $_F2$ van t alleen. Achter ‘&where’ staat het stelsel gewone differentiaalvergelijkingen waaraan deze functies moeten voldoen. De eigenwaarde is $_c1$ genoemd, en we hebben hem in de voorbeeldsessie vervangen door $-\lambda$. De mogelijke waarden van λ volgen uit de randvoorwaarden.

`build`

In de bibliotheek `PDEtools` zit de procedure `build` waarmee de algemene oplossing van de bovengenoemde gewone differentiaalvergelijkingen wordt bepaald. \diamond

In de tweede plaats leent vergelijking (24.21) zich om, door middel van tijd-discretisatie, *numerieke benaderingen* van de oplossing te berekenen.

Voorbeeldsessie

```
> restart; with(plots):
> PDV := diff(u(x,t),t) = k*diff(u(x,t),x$2):
Numerieke oplossing:
> L := Pi: k:= 1:
> f := x -> piecewise(x<0, 0, x<=L/2, x, x<L, L-x, 0 ):
> OPL := pdsolve( PDV, {u(0,t)=0, u(L,t)=0, u(x,0)=f(x)},
    numeric, u, time=t, range=0..L );

    OPL := module()
    export plot, plot3d, animate, value, settings;
    ...
    end module
> OPL:-animate(t=3,frames=100);
> plotlijst := seq(
    OPL:-plot(t=i,color=black,thickness=3), i=0..3, 0.5 );
> display(plotlijst);
```


(zie figuur 52.)

Toelichting

`numeric`

Als we de optie `numeric` aan de `pdsolve`-opdracht meegeven, dan moeten we uiteraard ook de randvoorwaarden en de beginvoorwaarde vermelden.

Het resultaat is een *module* (zie verder, §29.6), in feite een soort bibliotheek met de naam `OPL`, die de procedures `plot`, `plot3d`, enzovoort bevat. Met `OPL:-animate(...)` wordt bedoeld: ‘gebruik



FIGUUR 52. Numerieke benadering van de oplossing van de warmtevergelijking, voor verschillende waarden van t .

spacestep

de procedure `animate` uit de module `OPL`. Als voorbeeld hebben we in figuur 52 hetzelfde plaatje proberen te maken als figuur 51. We zien dat deze numerieke benadering uiteraard beter is bij $t = 0$ (want exact), maar dat er bij grotere waarden van t een zekere instabiliteit lijkt op te treden. Dit kan verbeterd worden door de `spacestep` kleiner te kiezen. Met `OPL:-settings()` krijgt men te zien welke `spacestep` Maple heeft gekozen. \diamond

Opgave 24.1

Voor ieder van de volgende partiële differentiaalvergelijkingen, (a) los de karakteristieke vergelijking op en schets de baan van sommige karakteristieken, (b) vereenvoudig de vergelijking met behulp van de karakteristieken, (c) bepaal de algemene oplossing van de getransformeerde vergelijking en (d) bepaal de algemene oplossing van de gegeven vergelijking.

- (a) $3u_x + 5u_y - xyu = 0$ (b) $u_x - u_y + yu = 0$
 (c) $u_x + 4u_y - xu = x$ (d) $x^2u_x + xyu_y + xu = x - y$
 (e) $u_x - y^2u_y - yu = 0$

Opgave 24.2

Classificeer de onderstaande partiële differentiaalvergelijkingen, schets sommige karakteristieken, en bepaal de canonieke vorm van de vergelijking.

- (a) $u_{xx} - 8u_{xy} + 2u_{yy} + xu_x - yu_y = 0$
 (b) $4u_{xx} + 2u_{xy} + u_{yy} - u_x + xyu_y + u = 0$
 (c) $3u_{xx} + 2u_{xy} - u_{yy} + yu_x - u_y = 0$
 (d) $2u_{xx} - 4u_{xy} + 2u_{yy} - y^2u_x + yu_y - xu = 0$

Opgave 24.3

Bepaal de oplossing van (24.12) met $\varphi(x) = \cos(3x)$, $\psi(x) = x$ en $c = 7$.

Opgave 24.4

Maak een animatie van (24.12) met $c = 1$, en als begincondities $\psi \equiv 0$ en

$$\varphi(x) = \begin{cases} \sin(x) & \text{als } -\pi \leq x \leq \pi \\ 0 & \text{als } |x| > \pi \end{cases}$$

op het tijdsinterval $[0, 5]$. Op welk tijdstip is de golf geheel uiteengevallen in twee delen, waarvan één deel naar links loopt en het andere naar rechts?

Opgave 24.5

Bepaal de oplossing van (24.15) als verder gegeven is dat $u(x, 0) = 1 - \exp(-x)$, $u_t(x, 0) = \cos(x)$ voor $x \geq 0$, en $u(0, t) = 0$. Maak een animatie van de oplossing.

Opgave 24.6

Bepaal de oplossing van (24.19) met de volgende begin- en randwaarden: $\varphi(x) = x$, $\psi(x) = e^{-x}$, $f(t) = t^2$, en $c = 1$.

Opgave 24.7

Modelleer de temperatuurverdeling in een staaf met geïsoleerde uiteinden door:

$$\begin{aligned} u_t &= k u_{xx} && \text{als } 0 < x < L, \quad t > 0 \\ u_t(0, t) &= u_t(L, t) = 0 && \text{als } t \geq 0 \\ u(x, 0) &= f(x) && \text{als } 0 \leq x \leq L. \end{aligned}$$

Kies $L = \pi$, $k = 4$, en

$$f(x) = \begin{cases} 0 & \text{als } 0 \leq x \leq \pi/2 \\ 50 & \text{als } \pi/2 \leq x \leq \pi. \end{cases}$$

Bereken een benadering van u met drie Fourier-termen en maak een animatie.